SIMULATION EXERCISES WITH SPICE – PART 1

Dr. J. E. Rayas-Sánchez

1. Simulate the following RLC parallel resonator




```
RLC Tank
* -----------------------------------------------------
* Dr. J.E. Rayas-Sanchez              February 24, 2016
* -----------------------------------------------------
*                         RLC Tank

Is  0  s  DC 0A AC 1mA PULSE(0A 10mA 0s 1ns 1ns 10ns 1.3us)
L   s  0  10uH
R   s  0  820
C   s  0  2.53nF

.control
TRAN 1ns 15us
plot v(s)
.endc

.end
```

After simulation:



Modifying the netlist to measure the input pulse current:



Increasing *R* to 820KΩ, and synchronizing the pulse period with the resonant frequency:





2. Apply special transient waveforms to a simple resistor



First apply a sinusoidal input signal with 1 V amplitude, 1 KHz frequency and 3 V offset:



Add a time delay of 1 ms and a damping factor of 500:



Switch to an exponential pulse input signal, such that you obtain: (τ_1 = 0.3 ms and τ_2 = 0.1 ms)



Switch to an FM input signal (F_c = 10 KHz , M = 7, F_s = 1 KHz):



Finally, implement the following piece-wise linear transient input signal:



3. Perform a transient analysis of the following series clipper




```
Series Clipper
* -----------------------------------------------------
* Dr. J.E. Rayas-Sanchez              February 24, 2016
* -----------------------------------------------------
*                       Series Clipper

Vs   in  0     DC 0V   AC 1V   SIN(0V 4V 1KHz)
Rs   in  in2   1
D1   in2 out   D1N4004
VR   out out2  DC  2V
RL   out2 0    100

* 1N4004 - 1A 400V General Purpose Rectifier
* Fairchild (now National Semiconductors)
.MODEL  D1N4004  D (IS=3.699E-09 RS=1.756E-02 N=1.774
+       XTI=3.0 EG=1.110 CJO=1.732E-11 M=0.3353
+       VJ=0.3905 FC=0.5 BV=400 IBV=1.0E-03)

* Quasi Ideal Diode
.model ideal_diode D (Is=1pA n=0.01)

.TRAN 10E-6 4E-3
.plot tran v(in) v(out)
.end
```

Using quasi-ideal diode:



Using 1N4004:



4. Perform an AC analysis on the RLC parallel resonator in exercise 1 from 50 Hz to 3 MHz, to obtain the following plot (use $R = 820\ \Omega$ and an AC source with 1 mA amplitude):





5. Simulate the following Common Base Amplifier




```
Common Base Amplifier
* -----------------------------------------------------
* Dr. J.E. Rayas-Sanchez                  March 25, 2006
* -----------------------------------------------------
*                   Common Base Amplifier

Vs    in  0    DC 0V    AC 1V
Vcc   vp  0    DC 12V
Q1    c   b    e  Q2N2222
*Q1    c   b    e  npn_ideal
RS    in  s    50
RE    e   0    910
RC    c   vp   270
R1    b   vp   5.3K
R2    b   0    7.5K
RL    out 0    1K
CE    e   s    10uF
CB    b   0    10uF
CL    c   out 10uF

.MODEL Q2N2222 NPN
+(IS=3.108E-15 XTI=3 EG=1.11 VAF=131.5 BF=217.5
+ NE=1.541 ISE=190.7E-15 IKF=1.296 XTB=1.5 BR=6.18
+ NC=2 ISC=0 IKR=0 RC=1 CJC=14.57E-12 VJC=.75
+ MJC=.3333 FC=.5 CJE=26.08E-12 VJE=.75 MJE=.3333
+ TR=51.35E-9 TF=451E-12 ITF=.1 VTF=10 XTF=2)

* Quasi ideal transistors
.model npn_ideal npn (Is=1.8fA Bf=150 VAf=300V)
.model pnp_ideal pnp (Is=1.8fA Bf=150 VAf=300V)

.control
AC DEC 10 10Hz 900MEGHz
plot vdb(out)
.endc
.end
```

Using quasi-ideal transistor:



Using 2N2222:



Insert SPICE commands to plot the gain in V/V (not in dB):

Using quasi-ideal transistor:



Using 2N2222:



Insert SPICE commands to plot the input impedance:

Using quasi-ideal transistor:



Using 2N2222:



Add SPICE commands to perform a transient analysis using a small amplitude sinusoidal input signal (10mV, 1KHz):

Using quasi-ideal transistor:



Using 2N2222:



Now make larger the amplitude of the sinusoidal input signal (2V, 1KHz):

Using quasi-ideal transistor:



Using 2N2222:



Add to the netlist an operating point SPICE command (.OP) and a transfer function SPICE command (.TF) and check the results: (to show the results in the command windows, you have to add "print all" in the control block)



Why are the transfer function results different to those obtained in the previous AC analyses?